
ЗАДАНИЕ

В лабораторной работе требуется написать программу моделирования буферной памяти со сквозной записью, моделирующую три обращения к памяти.

БЛОК-СХЕМА МОДЕЛИРОВАНИЯ РАБОТЫ СЕЛЕКТОРНОГО БУФЕРА СО СКОВЗНОЙ ЗАПИСЬЮ

КОД ПРОГРАММЫ НА ЯЗЫКЕ MCL
writeln " *******************************************************************";

writeln " *   model' gruppo-assoziativnoi KES pamati s priznakovim obmenom  *";

writeln " *******************************************************************";

{                  rabota modeliruetsa na urovne blokov                       }

{**************************** opicanie tipov *********************************}

{ tipi ispol'zuemie p/p }

mem op[32][8];   { operativnaa pamat' 32 slova po 8 bit }

mem bp[16][11];   { bufernaa pamat' 16 slov po 11 bit }

mem ispadr[3][5];  { pamat' ispolnitel'nih adresov, soderjit tri

                     ispolnitel'nih adresa }

mem inputs[3][8]; { cisla dla zapisi, soderjit cisla zapisivaemie v

                    pamat' }

mem coper[3][1]; { pamat' kodov operazii 1-ctenie, 0-zapis' }

mem rep_seg[3][2]; { nomer segmenta vitesnaemogo bloka, soderjit

              nomera segmentov bloki kotorih vitesnaytsa iz BP pri zamesenii }

struct iadr[5]      { adres aceiki OP 32 clova }

       blok[3],     { nomer bloka 8 st. }

       seg [2];     { nomer segmenta 4 st.}

struct bpadr[4]     { adres aceiki BP 16 clov }

       blok[3],     { nomer bloka 8st. }

       sect[1];     { nomer sektora 2st. }

struct buf_reg[10] { struktura slova bufernoi pamati,  v etu strukturu

scitivaetsa  slovo  BP pri lyboi obrabotke,  cto pozvolaet razlicat' v

nem  adresnuy cast' i dannie }

       data[8],     { dannie }

       address[2];  { adresnaa cast' }

struct subiadr[5]   { permennaa ispol'zuetsa v podprogrammah }

       blok[3],

       seg [2];

struct subbpadr[4]  { permennaa ispol'zuetsa v podprogrammah }

       blok[3],

       sect[1];

struct subbuf_reg[10]  { permennaa ispol'zuetsa v podprogrammah }

       data[8],

       address[2];

var i,j,     { vspomogatel'naa peremennaa }

    choice,  { nomer varianta }

    { peremennie ispol'zuemie v programme modelirovania }

    k,l,     { vspomogatel'naa peremennaa }

    op_code, { kod operazii }

    data,    { dannie }

    replace, { nomer segmenta vitesnaemogo bloka }

    s_no;    { nomer sektora, soderjasego neobhodimii blok }

{**************************** nacalo programmi *******************************}

gosub variant; { zadanie nacal'nih znacenii vseh peremennih }

for l:=0 to 2;

 replace:=rep_seg[l]; { scitivanie dannih iz massivov v tekusie peremennie }

 op_code:=coper[l];

 iadr:=ispadr[l];

 writeln "Ispolnitel'nii adres - ",$d2 iadr;

 data:=inputs[l];

 s_no:=255;            { nacal'noe zadanie }

 bpadr.blok:=iadr.blok;   { videlenie blokov s nujnim nomerom}

    { Poisk nalicia bloka v BP }

 bpadr.sect:=0;

 buf_reg:=bp[bpadr];

 if buf_reg.address=iadr.seg then s_no:=0;

 bpadr.sect:=1;

 buf_reg:=bp[bpadr];

 if buf_reg.address=iadr.seg then s_no:=1;

 if s_no<>255 then goto get_blok;      { Esli blok prisutstvuet v BP... }

    { Esli bloka net }

 write "Blok ",$d2 iadr.blok;

 writeln " ,segmenta ",$d2 iadr.seg," otsutstvuet v bufernoi pamati";

 writeln "Vitesnenie bloka prinadlejasego segmentu ",$d2 replace;

    { Poisk vitesnaemogo bloka }

 bpadr.sect:=0;

 buf_reg:=bp[bpadr];

 if buf_reg.address=replace then s_no:=0;

 bpadr.sect:=1;

 buf_reg:=bp[bpadr];

 if buf_reg.address=replace then s_no:=1;

    { Vitesnenie segmenta }

 bpadr.sect:=s_no;                       { nahojdenie adresa }

 buf_reg:=bp[bpadr];                     { scitivanie v bufernii registr }

 buf_reg.address:=iadr.seg; { zanesenie nomera segmenta v adresnuy cast' }

 buf_reg.data:=op[iadr]; { zapis' adresuemogo bloka iz OP }

 bp[bpadr]:=buf_reg;

get_blok:

 bpadr.sect:=s_no;                       { nahojdenie adresa }

 buf_reg:=bp[bpadr]; { ctenie bloka iz BP v bufernii registr }

    { vipolnenie operazii po kodu }

 if op_code=1 then goto read_data; { esli 1 to ctenie ... }

 { esli 0 to zapis' }

 k:=iadr;

 iadr.seg:=buf_reg.address;

 op[iadr]:=buf_reg.data;

 write "Perezapis' bloka ",$d2 iadr.blok;

 writeln " segmenta ",$d2 iadr.seg," v OP";

 iadr:=k;

 buf_reg.data:=data;

 writeln "Zapisani dannie ",$d4 data;

 goto end_job;

read_data:

 data:=buf_reg.data;

 writeln "Procitani dannie ",$d4 data;

end_job:

 bp[bpadr]:=buf_reg; { zapis' iz registra v bufernuy pamat' }
 op[iadr]:=buf_reg.data; { zapis' iz registra v OP}
 gosub print_bp;   { pecat' soderjimogo BP }

 gosub print_op;   { pecat' soderjimogo OP }

next;

end;

{**************************** konez programmi ********************************}

{**************************** podprogrammi ***********************************}

{ p/p zadanie varianta ,ispol'zuet p/p first,second,third }

variant:

{ zadanie ishodnih dannih po variantam }

read " Vvedite nomer varianta : ",choice;

if choice<=5 then gosub first;

if ((choice>5) and (choice<=10)) then gosub second;

if choice>10 then gosub third;

return; { variant }

{ p/p pervii variant }

first:

{ prisvaivanie nacal'nih znacenii aceikam OP }

for i:=0 to 31;

    op[i]:=(choice+2)*2+i-2;

next;

{ prisvaivanie nacal'nih znacenii aceikam BP }

for i:=0 to 7;

   buf_reg.data:=(choice+2)*2+i-2;

   buf_reg.address:=0;

   bp[i]:=buf_reg;

next;

for i:=8 to 15;

   buf_reg.data:=(choice+2)*2+i-2;

   buf_reg.address:=1;

   bp[i]:=buf_reg;

next;

{ naznacenie ispolnitel'nih adresov }

ispadr[0]:=4;

ispadr[1]:=28;

ispadr[2]:=22;

{ zadanie cisel dla zapisi }

for i:=0 to 2;

   inputs[i]:=i+8;

next;

{ KOPi }

coper[0]:=0;

coper[1]:=1;

coper[2]:=0;

{ nomera zamesaemih blokov }

rep_seg[0]:=0;

rep_seg[1]:=0;

rep_seg[2]:=1;

gosub print_bp;   { pecat' soderjimogo BP }

gosub print_op;   { pecat' soderjimogo OP }

gosub print;  { pecat' cisel dla zapisi i zamesaemih segmentov }

return; { first }

{ p/p vtoroi variant }

second:

{ prisvaivanie nacal'nih znacenii aceikam OP }

for i:=0 to 31;

    op[i]:=(choice+2)*2+i-8;

next;

{ prisvaivanie nacal'nih znacenii aceikam BP }

for i:=0 to 7;

   buf_reg.data:=(choice+2)*2+i;

   buf_reg.address:=1;

   bp[i]:=buf_reg;

next;

for i:=8 to 15;

   buf_reg.data:=(choice+2)*2+i;

   buf_reg.address:=2;

   bp[i]:=buf_reg;

next;

{ naznacenie ispolnitel'nih adresov }

ispadr[0]:=10;

ispadr[1]:=30;

ispadr[2]:=6;

{ zadanie cisel dla zapisi }

for i:=0 to 2;

   inputs[i]:=i+12;

next;

{ KOPi }

coper[0]:=1;

coper[1]:=0;

coper[2]:=1;

{ nomera zamesaemih blokov }

rep_seg[0]:=1;

rep_seg[1]:=2;

rep_seg[2]:=3;

gosub print_bp;   { pecat' soderjimogo BP }

gosub print_op;   { pecat' soderjimogo OP }

gosub print;  { pecat' cisel dla zapisi i zamesaemih segmentov }

return; { second }

{ p/p tretii variant }

third:

{ prisvaivanie nacal'nih znacenii aceikam OP }

for i:=0 to 31;

    op[i]:=(choice+2)/2+i+5;

next;

{ prisvaivanie nacal'nih znacenii aceikam BP }

for i:=0 to 7;

   buf_reg.data:=(choice+2)/2+i+21;

   buf_reg.address:=2;

   bp[i]:=buf_reg;

next;

for i:=8 to 15;

   buf_reg.data:=(choice+2)/2+i+21;

   buf_reg.address:=3;

   bp[i]:=buf_reg;

next;

{ naznacenie ispolnitel'nih adresov }

ispadr[0]:=8;

ispadr[1]:=26;

ispadr[2]:=10;

{ zadanie cisel dla zapisi }

for i:=0 to 2;

   inputs[i]:=i+23;

next;

{ KOPi }

coper[0]:=0;

coper[1]:=0;

coper[2]:=1;

{ nomera zamesaemih segmentov }

rep_seg[0]:=2;

rep_seg[1]:=2;

rep_seg[2]:=3;

gosub print_bp;   { pecat' soderjimogo BP }

gosub print_op;   { pecat' soderjimogo OP }

gosub print;  { pecat' cisel dla zapisi i zamesaemih segmentov }

return; { third }

{ p/p pecati soderjimogo bufernoi pamati }

print_bp:

   writeln "Soderjimoe bufernoi pamati :";

   writeln "      Cektor 0            Sektor 1";

   writeln "Blok  Seg. Izm. Dannie    Seg. Izm. Dannie";

   for i:=0 to 7;

      write " ",$d2 i;

      subbpadr.sect:=0;

      subbpadr.blok:=i;

      subbuf_reg:=bp[subbpadr];

      gosub print_br;

      subbpadr.sect:=1;

      subbuf_reg:=bp[subbpadr];

      gosub print_br;

      writeln;

   next;

return; { print_bp }

{ p/p pecati soderjimogo bufernogo registra }

print_br:

   write "    ", $d1 subbuf_reg.address;

   write "   ", $d4 subbuf_reg.data,"   ";

return; { print_br }

{ p/p pecati soderjimogo operativnoi pamati }

print_op:

   writeln "Soderjimoe operativnoi pamati:";

   writeln "    Blok #      0   1   2   3   4   5   6   7";

   for i:=0 to 3;

      write " Cegment #",$d1 i,"  ";

      subiadr.seg:=i;

      for j:=0 to 7;

         subiadr.blok:=j;

         write $d4 op[subiadr];

      next;

      writeln;

   next;

return; { print_op }

{ p/p pecati zapisivaemih cisel i zamesaemih segmentov }

print:

 writeln "------------------------------------";

 writeln "! N obras. ! Cisla dla ! N vitesn. !";

 writeln "!          ! zapisi    ! segmenta  !";

 writeln "------------------------------------";

 writeln "!     1    !",$d7 inputs[0],"    !",$d6 rep_seg[0],"     !";

 writeln "!     2    !",$d7 inputs[1],"    !",$d6 rep_seg[1],"     !";

 writeln "!     3    !",$d7 inputs[2],"    !",$d6 rep_seg[2],"     !";

 writeln "------------------------------------";

return; { print }
РЕЗУЛЬТАТ РАБОТЫ
*******************************************************************

 *   MODEL' GRUPPO-ASSOZIATIVNOI KES PAMATI S SKVOZNOY ZAPISYU     *

 *******************************************************************

 VVEDITE NOMER VARIANTA :

2

SODERJIMOE BUFERNOI PAMATI :

      CEKTOR 0            SEKTOR 1

BLOK  SEG. IZM. DANNIE    SEG. IZM. DANNIE

  0    0      6       1     14

  1    0      7       1     15

  2    0      8       1     16

  3    0      9       1     17

  4    0     10       1     18

  5    0     11       1     19

  6    0     12       1     20

  7    0     13       1     21

SODERJIMOE OPERATIVNOI PAMATI:

    BLOK #      0   1   2   3   4   5   6   7

 CEGMENT #0     6   7   8   9  10  11  12  13

 CEGMENT #1    14  15  16  17  18  19  20  21

 CEGMENT #2    22  23  24  25  26  27  28  29

 CEGMENT #3    30  31  32  33  34  35  36  37

------------------------------------

! N OBRAS. ! CISLA DLA ! N VITESN. !

!          ! ZAPISI    ! SEGMENTA  !

------------------------------------

!     1    !      8    !     0     !

!     2    !      9    !     0     !

!     3    !     10    !     1     !

------------------------------------

ISPOLNITEL'NII ADRES -  4

PEREZAPIS' BLOKA  4 SEGMENTA  0 V OP

ZAPISANI DANNIE    8

SODERJIMOE BUFERNOI PAMATI :

      CEKTOR 0            SEKTOR 1

BLOK  SEG. IZM. DANNIE    SEG. IZM. DANNIE

  0    0      6       1     14

  1    0      7       1     15

  2    0      8       1     16

  3    0      9       1     17

  4    0      8       1     18

  5    0     11       1     19

  6    0     12       1     20

  7    0     13       1     21

SODERJIMOE OPERATIVNOI PAMATI:

    BLOK #      0   1   2   3   4   5   6   7

 CEGMENT #0     6   7   8   9   8  11  12  13

 CEGMENT #1    14  15  16  17  18  19  20  21

 CEGMENT #2    22  23  24  25  26  27  28  29

 CEGMENT #3    30  31  32  33  34  35  36  37

ISPOLNITEL'NII ADRES - 28

BLOK  4 ,SEGMENTA  3 OTSUTSTVUET V BUFERNOI PAMATI

VITESNENIE BLOKA PRINADLEJASEGO SEGMENTU  0

PROCITANI DANNIE   34

SODERJIMOE BUFERNOI PAMATI :

      CEKTOR 0            SEKTOR 1

BLOK  SEG. IZM. DANNIE    SEG. IZM. DANNIE

  0    0      6       1     14

  1    0      7       1     15

  2    0      8       1     16

  3    0      9       1     17

  4    3     34       1     18

  5    0     11       1     19

  6    0     12       1     20

  7    0     13       1     21

SODERJIMOE OPERATIVNOI PAMATI:

    BLOK #      0   1   2   3   4   5   6   7

 CEGMENT #0     6   7   8   9   8  11  12  13

 CEGMENT #1    14  15  16  17  18  19  20  21

 CEGMENT #2    22  23  24  25  26  27  28  29

 CEGMENT #3    30  31  32  33  34  35  36  37

ISPOLNITEL'NII ADRES - 22

BLOK  6 ,SEGMENTA  2 OTSUTSTVUET V BUFERNOI PAMATI

VITESNENIE BLOKA PRINADLEJASEGO SEGMENTU  1

PEREZAPIS' BLOKA  6 SEGMENTA  2 V OP

ZAPISANI DANNIE   10

SODERJIMOE BUFERNOI PAMATI :

      CEKTOR 0            SEKTOR 1

BLOK  SEG. IZM. DANNIE    SEG. IZM. DANNIE

  0    0      6       1     14

  1    0      7       1     15

  2    0      8       1     16

  3    0      9       1     17

  4    3     34       1     18

  5    0     11       1     19

  6    0     12       2     10

  7    0     13       1     21

SODERJIMOE OPERATIVNOI PAMATI:

    BLOK #      0   1   2   3   4   5   6   7

 CEGMENT #0     6   7   8   9   8  11  12  13

 CEGMENT #1    14  15  16  17  18  19  20  21

 CEGMENT #2    22  23  24  25  26  27  10  29

 CEGMENT #3    30  31  32  33  34  35  36  37
начало





Запись байта в ОП и БП по ИА;


БитПрисутствия:=1





Чтение в ЦП из БП по ИА





да





нет





Код операции чтение?





Запись блока из ОП в БП по ИА.;


БитПрисутствия:=1





да





нет





Блок присутствует в БП?





Вытеснение из БП заданного сегмента





да





нет





начало





нет





да





Сегмент присутствует в БП?





Есть обращение к памяти?








